
*                                 R2  12k          R3 500
*                         ____/\  /\  /\_______/\  /\  /\_
*     IABC0  = .3mA->0   |      \/  \/       |   \/  \/  _|_
*                        |    INN1           |           \\\
*    ____|\     OUT0 |\  |_______|\          |______|\       OUT2
*  _|_   | \/\/\_____| \_|       | \/\/\__          | \/\/\_____
*  \\\ __| /\/\/ _|_ | /       __| /\/\/  |       __| /\/\/     |
*     |  |/      ___ |/200PF   | |/       |     _|_ |/          |
*     | B_OTA0   _|_           | B_OTA1   |     \\\ B_OTA2      |
*     |          \\\ C0        |          |                     |
*     |________________________|__________|        _/\  /\  /\__|
*                     |    INP1                  _|_  \/  \/
*        _/\  /\  /\__|                          \\\   R4    5K
*      _|_  \/  \/      R1   IABC1=500uA0
*      \\\          5K                          IABC1=500uA0
*

.OPTIONS  GMIN=1e-18       METHOD=trap   srcsteps = 1  gminsteps = 1 trtol=1 VNTOL=1m
V_Iabc0   VIABC0     0     PWL    ( 0 .3m  1m 0 )  
B_OTA0    OUT0       0     I =   -1*v(VIABC0)*tanh( v(INP1)/.052)
C0        OUT0       0     2000p   IC=1
B_BUF0    INN1       0     V =    v(OUT0)
V_Iabc1   VIABC1     0     DC .5m
B_OTA1    INP1       0     I =   -1*v(VIABC1)*tanh( (v(INP1)-v(INN1)) /.052)
R1        INP1       0     5000
B_BUF1    VOUT1      0     V =    v(INP1)
Cstray    INP1       0     15p
R2        INN1       INN2  12k
R3        INN2       0     500
B_OTA2    OUT2       0     I =   -1*v(VIABC1)*tanh(-v(INN2) /.052)
R4        OUT2       0     5K
.tran     .1u        1m    0      .1u   UIC

.control
run
set pensize = 2
plot  v(inp1) v(out0)  v(out2)
.endc

.end

.end

=====================END_OF_SPICE============================

This code works with winspice.

No Function generator is complete with out a a sinewave
output. It may not have low distortion, but putting a
triangle wave form at the right magnitude into a standard
OTA will generate something that looks close to a sine
wave.   

It would be something too build such a function generator
today in BiCMOS. Transistors are far smaller and far faster
and have almost twelve orders of magnitude of dynamic range
compare to the six orders of magnitude for the LM13600. 
Both AM and FM are available. And all this could be



done completely inside silicon. 


